
1 [bookmark: _Toc122434485][bookmark: _Toc134086713]Overview
This document lists all the essential changes needed to correct problems in the iwd-TTCN3-B2022-09_D23wk12 related to the title of this CR.
Contact:	Parikshit Bhise
	parikshit.bhise@rohde-schwarz.com 

2 [bookmark: _Toc134086714]Corrections required
2.1 [bookmark: _Toc134086715][bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665]Function fl_TC_8_1_4_3_2And5_TestBody
	Function name
	fl_TC_8_1_4_3_2And5_TestBody

	Reason for change
	1. At step 3, the UE may send the Reestablishment Req before the expiry of T304 timer as soon as the NR Cell 2 is turned off at T1 time instance. This could fail the test case because the reestablishment req sent by the UE at this step would be because of radio link failure on NR Cell 2 and not because of the T304 timer expiry. Thus, UE would not get the chance to execute TS 38.331, clause 5.3.5.8.3 on T304 of the MCG expires to act acorrdingly on sending Failure Information. A prose CR is also raised for this correction.
2. In current TTCN implementation at STEP 2 SRB1 and SRB2 are released and reconfigured back to cater the need of TS 38.331, 5.3.5.8.3: the UE shall, for each SRB, re-establish the RLC entity for the source PCell. But, SS will not retain the DRBs configuration on source Cell if its SRBs are already released. Therefore, for completeness DRBs should also be released and reconfigured along with the SRBs and PDCP count should also be provided for all again.
3. C-RNTI based contention resolution configurations are also lost due
to SRBs release method of TTCN to achive re-establish the RLC entity which is also expected to intact as it is not the purpose here.
4. Security keys are lost with SRBs and DRB release.
 

	Summary of change
	1. Keep NR cell 2 “suitableIntraFreq” level (-94dbM) instead of “OFF” as per T0. This way, the UE will always find NR Cell 2 as suitable cell but SS would not respond as per Table 8.1.4.3.2.3.2-3: Parallel behaviour. Finally when T304 gets expire UE would get a chance to act according to TS 38.331, clause 5.3.5.8.3 on T304 of the MCG expires and to send FailureInformation based on TS 38.331, section 7.1- T304 expiry - If any DAPS bearer is configured and if there is no RLF in source PCell, initiate the failure information procedure. In order to ensure this happens as per prose new TTCN changes are added to execute T304 expiry and to ensure UE tries to RACH on target cell during this time frame and moved all configuration changes once it gets expire to meet the prose condition for the further test behavior. Also, added checks to capture RRCReestablishmentRequest in case UE behaviour is non-compliant with prose while T304 is running.
2. Stored PDCP count for SRBs and DRB before releasing all and reapplied once configured back including the DRB.
3. Configured CBRA CRNTI Based RACH procedure to intact all configuration as is.
4. Reapplied Security keys including DRB.


	TTCN module
	\ NR5GC\8_1_4\RRC_DAPS_Handover_NR5GC.ttcn

	MCC160 Comment
	



Before Change:
	function fl_TC_8_1_4_3_2And5_TestBody(NR_CellId_Type p_SourceCellId,
                                        NR_CellId_Type p_TargetCellId) runs on NR5GC_PTC
  {
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;
    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;
    var template (value) DL_DCCH_Message v_RRCReconfiguration;
    var SubFrameTiming_Type v_Timing;
    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList1; //@sic R5s221242 sic@
    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList2; //@sic R5s221242 sic@
    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList; //@sic R5s221242 sic@
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_DRB; //@sic R5s230362 sic@
    var NR_SecurityParams_Type v_RRC_SecurityParams;
    timer t_Wait := 5.0;
 
    v_CellPowerList_AtT0 := {cs_NR_CellPower(p_SourceCellId, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // TC n/a for FR2
                             cs_NR_CellPower(p_TargetCellId, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT1 := {cs_NR_CellPower(p_TargetCellId, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT2 := {cs_NR_CellPower(p_TargetCellId, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT3 := {cs_NR_CellPower(p_SourceCellId, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    //Set power level to T0
    f_NR_SetCellPowerList (v_CellPowerList_AtT0);

    // Configuration of Target Cell for no RACH response transmission)
    f_NR_SS_RachProcedureConfig(p_TargetCellId, cs_NR_RachProcedureConfig_NoResponse, cs_TimingInfo_Now);

    //@siclog "Step 1" siclog@
    // The SS transmits an RRCReconfiguration message containing reconfigurationWithSync to order UE to perform DAPS handover to NR Cell 2.
    // DRB#n of the first PDU session is configured as DAPS bearer.
    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(p_SourceCellId, tsc_NR_BWP_Id);
    v_RACH_ConfigGeneric.preambleTransMax := n200; // Table 8.1.4.3.2.3.3-4
    v_RRCReconfiguration := f_NR5GC_RRCReconfiguration_IntraNR_HO_DAPS(p_TargetCellId,
                                                                       -,
                                                                       cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount),
                                                                       cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));
    
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, v_RRCReconfiguration));
    
    //@siclog "Step 2" siclog@
    // The SS changes the power level setting according to the row "T1".
    f_NR_SetCellPowerList (v_CellPowerList_AtT1);

    //Acc. 38.331, 5.3.5.8.3: the UE shall, for each SRB, re-establish the RLC entity for the source PCell. When we release an SRB in the SS, the PDCP count is lost, so we need to set it to the original value once the SRB is configured again.
    //@sic R5s221242 sic@
    v_PdcpCountInfoList1 := f_NR_SS_PdcpCount_Get(p_SourceCellId,tsc_NR_RbId_SRB1);
    v_PdcpCountInfoList2 := f_NR_SS_PdcpCount_Get(p_SourceCellId,tsc_NR_RbId_SRB2);
    v_PdcpCountInfoList := v_PdcpCountInfoList1 & v_PdcpCountInfoList2;
    f_NR_SS_SRBs_DRBs_Release(p_SourceCellId, cs_TimingInfo_Now, {}, tsc_NoCnfReq);
    f_NR_SS_SRBs_DRBs_Config(p_SourceCellId, cs_TimingInfo_Now, omit, tsc_NoCnfReq);

    //Reapply security keys for SRBs in the source cell - @sic R5s230387 sic@
    v_RRC_SecurityParams := f_NR_Security_Get();
    v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_SourceCellId, false, -, v_PdcpCountInfoList);
    f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_SourceCellId,
                                          v_RRC_SecurityParams.AS_Integrity,
                                          v_RRC_SecurityParams.AS_Ciphering);

    //@sic R5s221242, R5s230387 sic@
    f_NR_SS_PdcpCount_Set(p_SourceCellId, v_PdcpCountInfoList, cs_TimingInfo_Now);

    //@siclog "Step 3-4" siclog@
    // Check: Does the UE transmit an RRCReestablishmentRequest message on NR Cell 1 within 5s?
    // Check: Does the UE transmit a FailureInformation message with failureType-r16 set to daps-failure on NR Cell 1 within 5s?
    t_Wait.start;
    alt {
      [] t_Wait.timeout{}
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND (p_SourceCellId, cr_38508_RRCReestablishmentRequest(?, ?))){
        t_Wait.stop;
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "RRCReestablishmentRequest received on NR Cell 1");
      }
      [] SRB.receive(car_NR_SRB1_RrcPdu_IND (p_SourceCellId, cr_38508_FailureInformation(-, cr_FailureInformation_v1610({daps_failure})))){ //@sic R5s221242 sic@
        t_Wait.stop;
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4"); //@sic R5s221242 sic@
      }
    }
    
    //@siclog "Step 5" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath(p_SourceCellId, "Step 5");

    //@siclog "Step 6" siclog@
    // The SS changes the power level setting according to the row "T2".
    f_NR_SetCellPowerList (v_CellPowerList_AtT2);

    //@siclog "Step 7" siclog@
    //v_Timing := f_NR_GetNextSendOccasion_DL(p_SourceCellId, 700); // 700 ms; @sic R5s230320 sic@: v_Timing is an out parameter set inside f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig
    v_NR_PdcpCountInfoList_DRB := f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig(p_SourceCellId, p_TargetCellId, v_Timing); //@sic R5s230362 sic@
    
    f_Delay(0.8); //@sic R5s230362, R5s230387 sic@
    
    //@siclog "Step 8" siclog@
    // The SS changes the power level setting according to the row "T3" and waits 300ms to ensure T310 expires in NR Cell 1.
    f_NR_SetCellPowerList (v_CellPowerList_AtT3);
    f_Delay(0.3); //@sic R5s221242 sic@
   
    f_NR_SS_PdcpCount_Set(p_TargetCellId, v_NR_PdcpCountInfoList_DRB, cs_TimingInfo_Now);  //Set count for DAPS DRB; TS 38.523-3, 7.3.5.3.5, Step 10A @sic R5s230362 sic@
   
    //@siclog "Steps 9-10" siclog@
    // The SS transmits Random Access Response to respond to the latest preamble on NR Cell 2.
    // Check: Does UE transmit an RRCReconfigurationComplete message in NR Cell 2?
    f_NR5GC_DAPS_HO_RACHProcedure_RRCReconfigurationComplete(p_TargetCellId,
                                                             "Step 10",
                                                             -,
                                                             cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));
    //@siclog "Step 11" siclog@
    // Check: Does the test result of generic test procedure in TS 38.508-1 [4] Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB#n of the first PDU session on NR Cell 2?
    f_NR5GC_DAPS_HO_CheckDataPath(p_TargetCellId, "Step 11");

    //@siclog "Steps 12-13" siclog@
    // The SS transmits an RRCReconfiguration message with condition DAPS_HO_ReleaseSource in NR Cell 2.
    // Check: Does the UE transmits an RRCReconfigurationComplete message in NR Cell 2.
    f_NR5GC_DAPS_HO_SourceRelease(p_SourceCellId, p_TargetCellId, "Step 13", false); //@sic R5-230116, R5s230362 sic@
    f_NR5GC_DAPS_HO_SourceRelease_SSConfig(p_SourceCellId);
  }

}




After Change:
	function fl_TC_8_1_4_3_2And5_TestBody(NR_CellId_Type p_SourceCellId,
                                        NR_CellId_Type p_TargetCellId) runs on NR5GC_PTC
  {
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT0;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT1;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT2;
    var template (value) NR_CellPowerList_Type v_CellPowerList_AtT3;
    var template (value) RACH_ConfigGeneric v_RACH_ConfigGeneric;
    var template (value) DL_DCCH_Message v_RRCReconfiguration;
    var SubFrameTiming_Type v_Timing;
    //var NR_PdcpCountInfoList_Type v_PdcpCountInfoList1; //@sic R5s221242 sic@
    //var NR_PdcpCountInfoList_Type v_PdcpCountInfoList2; //@sic R5s221242 sic@
    //var NR_PdcpCountInfoList_Type v_PdcpCountInfoList; //@sic R5s221242 sic@
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList_DRB; //@sic R5s230362 sic@
    var NR_SecurityParams_Type v_RRC_SecurityParams;
    var IntegerList_Type v_DrbIdList ;           // WA#WA=1097036
    timer t_Wait := 5.0;
    var boolean v_T304_Timeout := false;         // WA#WA=1097036
    timer t_T304Min;                             // WA#WA=1097036
    var boolean v_Rcvd_RachPreambleInd := false; // WA#WA=1097036
    var NR_PdcpCountInfoList_Type v_NR_PdcpCountInfoList; // WA#WA=1097036
    var template (value) NR_DRBInfoList_Type v_DRBInfoList; //WA#WI=1097036
    var template (value) NR_RadioBearerList_Type v_RadioBearerList; // WA#WA=1097036
    var NR_PhysicalParameters_Type v_NR_PhysicalParameters; // WA#WA=1097036
    var template(value) NR_RachProcedureConfig_Type v_NR_RachProcedure; // WA#WA=1097036
      
    v_CellPowerList_AtT0 := {cs_NR_CellPower(p_SourceCellId, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE), // TC n/a for FR2
                             cs_NR_CellPower(p_TargetCellId, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT1 := {cs_NR_CellPower(p_TargetCellId, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT2 := {cs_NR_CellPower(p_TargetCellId, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    v_CellPowerList_AtT3 := {cs_NR_CellPower(p_SourceCellId, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)}; // TC n/a for FR2
    //Set power level to T0
    f_NR_SetCellPowerList (v_CellPowerList_AtT0);

    // Configuration of Target Cell for no RACH response transmission)
    f_NR_SS_RachProcedureConfig(p_TargetCellId, cs_NR_RachProcedureConfig_NoResponse, cs_TimingInfo_Now);
    f_NR_SS_SystemIndCtrlConfig(p_TargetCellId, cds_NR_SystemIndCtrl_RachPreamble(enable));  //WA#WA=1097036
    //@siclog "Step 1" siclog@
    // The SS transmits an RRCReconfiguration message containing reconfigurationWithSync to order UE to perform DAPS handover to NR Cell 2.
    // DRB#n of the first PDU session is configured as DAPS bearer.
    v_RACH_ConfigGeneric := f_NR_CellInfo_GetRACH_ConfigGeneric(p_SourceCellId, tsc_NR_BWP_Id);
    v_RACH_ConfigGeneric.preambleTransMax := n200; // Table 8.1.4.3.2.3.3-4
    v_RRCReconfiguration := f_NR5GC_RRCReconfiguration_IntraNR_HO_DAPS(p_TargetCellId,
                                                                       -,
                                                                       cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount),
                                                                       cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));
    
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourceCellId, cs_TimingInfo_Now, v_RRCReconfiguration));
    
    //@siclog "Step 2" siclog@
    // The SS changes the power level setting according to the row "T1".
    //f_NR_SetCellPowerList (v_CellPowerList_AtT1); WA#WA=1097036 - Need a Prose change.
      //WA#WA=1097036
   // EXCEPTION: In parallel with step 3 or Step 4, parallel behaviour defined in table 8.1.4.3.2.3.2-3 is executed repeatedly for the duration of T304 to check the if T304 of the MCG expires clause as per [TS 38.331, clause 5.3.5.8.3].
     t_Wait.start;
       //@siclog "Step 3-4" siclog@
    // Check: Does the UE transmit an RRCReestablishmentRequest message on NR Cell 1 within 5s?
    // Check: Does the UE transmit a FailureInformation message with failureType-r16 set to daps-failure on NR Cell 1 within 5s?
      
    t_T304Min.start(f_NR_SetTimerToleranceMin(p_TargetCellId, rrcTimer, 1.0)); // 1s timer
    while (not v_T304_Timeout) {
      alt {
        [] SYSIND.receive(car_NR_PRACH_Preamble_IND(p_TargetCellId, cr_TimingInfo_Any, ?))
           {
             v_Rcvd_RachPreambleInd := true;
           }
          
          [] SRB.receive(car_NR_SRB0_RrcPdu_IND (p_SourceCellId, cr_38508_RRCReestablishmentRequest(?, ?))){
        t_Wait.stop;
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "RRCReestablishmentRequest received on NR Cell 1");
      }
        [] t_T304Min.timeout
         {
             v_T304_Timeout := true;
             f_NR_SS_SystemIndCtrlConfig(p_TargetCellId, cds_NR_SystemIndCtrl_RachPreamble(disable));
                    
             v_NR_PdcpCountInfoList:=f_NR_SS_PdcpCount_Get(p_SourceCellId); //  Get PDCP count info for all SRBs and DRBs before releasing.
             v_DrbIdList := f_NR_GetActiveDRBs(p_SourceCellId); //WA#WI=1097036
    
            f_NR_SS_SRBs_DRBs_Release(p_SourceCellId, cs_TimingInfo_Now, v_DrbIdList, tsc_NoCnfReq); //WA#WI=1097036 - Add DRB release as well.
            
             // Configure SRBs and DRBs again not just the SRBs.
            v_DRBInfoList := f_NR5GC_MobileInfo_GetDRBInfoList();//WA#WI=1097036
            v_RadioBearerList := f_NR5GC_GetRadioBearerList(v_DRBInfoList);//WA#WI=1097036
            f_NR_SS_SRBs_DRBs_Config(p_SourceCellId, cs_TimingInfo_Now, v_RadioBearerList, tsc_NoCnfReq);//WA#WI=1097036

            //WA#WI=1097036 - Configure CBRA CRNTI Based RACH procedure again as it has been lost with SRBs release. 
             v_NR_PhysicalParameters := f_NR_CellInfo_GetPhysicalParameters(p_SourceCellId);
             v_NR_RachProcedure := cs_NR_RachProcedureConfig_CBRA_CrntiBased(v_NR_PhysicalParameters, -, -, -, cs_NR_AssignedBWPs_InitialBWP);
             f_NR_SS_RachProcedureConfig(p_SourceCellId, v_NR_RachProcedure);
    
            //WA#WI=1097036 - Reapply security keys.
            v_RRC_SecurityParams := f_NR_Security_Get();
            v_RRC_SecurityParams.AS_Ciphering.ActTimeList := f_NR_RRC_CipherActTime_GetCurrent(p_SourceCellId, false, -, v_NR_PdcpCountInfoList); //@sic R5s220123 sic@
            f_NR_SS_RRC_EnableIntProt_CiphULandDL(p_SourceCellId,
                                            v_RRC_SecurityParams.AS_Integrity,
                                            v_RRC_SecurityParams.AS_Ciphering);
     
             //@sic R5s221242, R5s230387 sic@
            f_NR_SS_PdcpCount_Set(p_SourceCellId, v_NR_PdcpCountInfoList, cs_TimingInfo_Now);
                
         }
      }
    }
    if (not v_Rcvd_RachPreambleInd){
       f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE did not send PRACH preamble on NR Cell 2");
    }

      // T304 of the MCG has expired.
    alt {
      [] t_Wait.timeout{}
      [] SRB.receive(car_NR_SRB0_RrcPdu_IND (p_SourceCellId, cr_38508_RRCReestablishmentRequest(?, ?))){
        t_Wait.stop;
        f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "RRCReestablishmentRequest received on NR Cell 1");
      }
      [] SRB.receive(car_NR_SRB1_RrcPdu_IND (p_SourceCellId, cr_38508_FailureInformation(-, cr_FailureInformation_v1610({daps_failure})))){ //@sic R5s221242 sic@
        t_Wait.stop;
        f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4"); //@sic R5s221242 sic@
      }
    }
    
    //@siclog "Step 5" siclog@
    f_NR5GC_DAPS_HO_CheckDataPath(p_SourceCellId, "Step 5");

    //@siclog "Step 6" siclog@
    // The SS changes the power level setting according to the row "T2".
    f_NR_SetCellPowerList (v_CellPowerList_AtT2);

    //@siclog "Step 7" siclog@
    //v_Timing := f_NR_GetNextSendOccasion_DL(p_SourceCellId, 700); // 700 ms; @sic R5s230320 sic@: v_Timing is an out parameter set inside f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig
    v_NR_PdcpCountInfoList_DRB := f_NR5GC_DAPS_HO_RRCReconfiguration_SSConfig(p_SourceCellId, p_TargetCellId, v_Timing); //@sic R5s230362 sic@
    
    f_Delay(0.8); //@sic R5s230362, R5s230387 sic@
    
    //@siclog "Step 8" siclog@
    // The SS changes the power level setting according to the row "T3" and waits 300ms to ensure T310 expires in NR Cell 1.
    f_NR_SetCellPowerList (v_CellPowerList_AtT3);
    f_Delay(0.3); //@sic R5s221242 sic@
   
    f_NR_SS_PdcpCount_Set(p_TargetCellId, v_NR_PdcpCountInfoList_DRB, cs_TimingInfo_Now);  //Set count for DAPS DRB; TS 38.523-3, 7.3.5.3.5, Step 10A @sic R5s230362 sic@
   
    //@siclog "Steps 9-10" siclog@
    // The SS transmits Random Access Response to respond to the latest preamble on NR Cell 2.
    // Check: Does UE transmit an RRCReconfigurationComplete message in NR Cell 2?
    f_NR5GC_DAPS_HO_RACHProcedure_RRCReconfigurationComplete(p_TargetCellId,
                                                             "Step 10",
                                                             -,
                                                             cs_NR_RACH_ConfigDedicated_Uplink(cs_38508_RACH_ConfigDedicated(v_RACH_ConfigGeneric, -, f_NR_CellInfo_GetSSB_Index(p_TargetCellId))));
    //@siclog "Step 11" siclog@
    // Check: Does the test result of generic test procedure in TS 38.508-1 [4] Table 4.9.1-1 indicate that the UE is capable of exchanging IP data on DRB#n of the first PDU session on NR Cell 2?
    f_NR5GC_DAPS_HO_CheckDataPath(p_TargetCellId, "Step 11");

    //@siclog "Steps 12-13" siclog@
    // The SS transmits an RRCReconfiguration message with condition DAPS_HO_ReleaseSource in NR Cell 2.
    // Check: Does the UE transmits an RRCReconfigurationComplete message in NR Cell 2.
    f_NR5GC_DAPS_HO_SourceRelease(p_SourceCellId, p_TargetCellId, "Step 13", false); //@sic R5-230116, R5s230362 sic@
    f_NR5GC_DAPS_HO_SourceRelease_SSConfig(p_SourceCellId);
  }

}




